
運動方程式の方法:
サブルーチンの使い方



サブルーチン(subroutine)

• 多くのプログラムは，小さい部分に分割出来
る

• 例えば運動方程式の方法の場合；
– 初期値を決める

– 時間発展を行う

– 結果を書く

• このとき，時間発展の部分はいろいろなプロ
グラムで同じ構造をしている；時間tの状態を
運動方程式に従ってt+δtにするだけ。



サブルーチンを使う理由

• スペースの節約

• デバッグがしやすい
一度書いて間違いがないことを確かめたらもうその
まま直さないでよい



関数とサブルーチン

• 関数：exp(x), sin(x)のように変数を与えてあ
る手続きで別の数を出すもの。Fortran90を
始め，ほとんどのプログラミング言語では，
我々が自分で関数を作ることが出来る。

• サブルーチン：変数に対してある手続きを行
い，別の変数を作るプロセス。変数は複数で
もよい。



関数プログラムの例

ここでは華氏から摂氏の変換プログラムを扱う。
摂氏＝(華氏-32)*5/9

（私は華氏68度＝摂氏20度として，それからのずれを5/9≒0.5
して，計算している。例えばニュースで華氏80度と言われたら
(80-32)*5/9=26.67をやらず，(80-68)/2+20=26と概算してる。）



program fahrenheit2celsius
!-------------------------
!  This is a program to convert F to C
!  2005/5/16  Written by T. Ohtsuki
!-------------------------
implicit none  ! Always begin with this statement
integer,parameter::double=selected_real_kind(14)
real(kind=double):: fahrenheit,celsius

!---------main program---------
print *, "Input temperature in Fahrenheit."
read *, fahrenheit

celsius=f2c(fahrenheit)

print *,"It is ",celsius,"degree in celsius"

contains
!------------------- Subroutines ---------------------

function f2c(temperature)
!-------------------------
!   function to convert fahrenheit to celsius
!-------------------------
IMPLICIT NONE
integer,parameter::double=selected_real_kind(14)
REAL(kind=double), INTENT(IN) :: temperature
REAL(kind=double):: f2c

f2c=(temperature-32._double)/1.8_double

END function f2c

end program fahrenheit2celsius

これをfahrenheit2celsius.f90としてsave



別のやり方：ファイルを分ける

program fahrenheit2celsius
!-------------------------
!  This is a program to convert F to C
!  2005/5/16  Written by T. Ohtsuki
!-------------------------
implicit none  ! Always begin with this statement
integer,parameter::double=selected_real_kind(14)
real(kind=double):: fahrenheit,celsius

interface
function f2c(temperature)

integer,parameter::double=selected_real_kind(14)
real(kind=double)::f2c
real(kind=double),intent(IN)::temperature

end function f2c
end interface

!---------main program---------
print *, "Input temperature in Fahrenheit."
read *, fahrenheit
celsius=f2c(fahrenheit)

print *,"It is ",celsius,"degree in celsius"

end program fahrenheit2celsius

これはfahrenheit2celsius2.f90としてsave

このプログラムではf2cの
変数の型がわからないので
このinterfaceで判断させる



関数を独立させる

function f2c(temperature)
!-------------------------
!   function to convert fahrenheit to celsius
!-------------------------
IMPLICIT NONE
integer,parameter::double=selected_real_kind(14)
REAL(kind=double), INTENT(IN) :: temperature
REAL(kind=double):: f2c !interfaceを使わない例

f2c=(temperature-32._double)/1.8_double

END function f2c

前のをfahrenheit2celsius2.f90, これをf2c.f90と名付ける。
f90 fahrenheit2celsius2.f90 f2c.f90で実行ファイルを作る。

これはf2c.f90としてsave



関数のモジュール化

• 似たような関数をまとめたい

– モジュール化

– 例えば華氏摂氏の逆で摂氏華氏を行う関数
も一つにしておきたい場合



program fahrenheit2celsius
!-------------------------
!  This is a program to convert F to C
!  2005/5/16  Written by T. Ohtsuki
!-------------------------
use temperaturelibs
implicit none  ! Always begin with this statement
integer,parameter::double=selected_real_kind(14)
real(kind=double):: fahrenheit,celsius

!---------main program---------
print *, "Input temperature in Fahrenheit."
read *, fahrenheit

celsius=f2c(fahrenheit)

print *,"It is ",celsius,"degree in celsius"

print *, "Input temperature in Celsius."
read *, celsius
fahrenheit=c2f(celsius)

print *,"It is ",fahrenheit,"degree in fahrenheit."

end program fahrenheit2celsius



モジュールとはこんなもの
module temperaturelibs
implicit none
contains

function f2c(temperature)
!-------------------------
!   function to convert fahrenheit to celsius
!-------------------------
IMPLICIT NONE
integer,parameter::double=selected_real_kind(14)
REAL(kind=double), INTENT(IN) :: temperature
REAL(kind=double):: f2c

f2c=(temperature-32._double)/1.8_double

END function f2c

function c2f(temperature)
!-------------------------
!   function to convert fahrenheit to celsius
!-------------------------
IMPLICIT NONE
integer,parameter::double=selected_real_kind(14)
REAL(kind=double), INTENT(IN) :: temperature
REAL(kind=double):: c2f

c2f=temperature*1.8_double+32

END function c2f

end module temperaturelibs



モジュールの形のプログラムのコ
ンパイルの仕方

• まず使うモジュールプログラム（ここでは
temperaturelibs.f90とする）を、あらかじめコンパイ
ルしておく。

• f90 -c temperaturelibs.f90   (-c はコンパイルだけ
やり、実行ファイルは作るなというオプション) する
と、temperaturelibs.o, temperaturelibs.modという
ファイルができる。

• 次に

f90 fahrenheit2celsius3.f90 temperaturelibs.o
というコマンドで実行ファイルを作る。



サブルーチンとは？

• 関数に似ている

• 関数よりも広く，手続きを行う。一群の変数を
いれると一群の変数を返してくれる

• 一つの関数の値だけでなく，例えば行列を変
数にすると，その固有値，固有ベクトルを返し
てくれるようなものがサブルーチン

• 具体的に，f2cをサブルーチンの形で書き直
そう。



program fahrenheit2celsius
!-------------------------
!  This is a program to convert F to C
!  2005/5/16  Written by T. Ohtsuki
!  2005/5/17  subroutine version
!-------------------------
implicit none  ! Always begin with this statement
integer,parameter::double=selected_real_kind(14)
real(kind=double):: fahrenheit,celsius

!---------main program---------
print *, "Input temperature in Fahrenheit."
read *, fahrenheit

call f2c(fahrenheit,celsius)

print *,"It is ",celsius,"degree in celsius"

end program fahrenheit2celsius

これはfahrenheit2celsius4.f90としてsave



subroutine f2c(fah,cel)
!-------------------------
!   function to convert fahrenheit to celsius
!-------------------------
IMPLICIT NONE
integer,parameter::double=selected_real_kind(14)
REAL(kind=double), INTENT(IN) :: fah
REAL(kind=double),INTENT(OUT):: cel

cel=(fah-32._double)/1.8_double

END subroutine f2c

こちらはf2csubroutine.f90としてsave



課題

• 中点法のプログラムをサブルーチン化する。

• 来週の授業までに出す。出来た人は，落下運
動のプログラムにこれを適用し，動作を確認
して，来週の出席メールに添付すること。

